

# Compiling away first-class functions: closure conversion and defunctionalization

MPRI FUN

François Pottier



2025–2026

# Program transformations

In the first week, we studied how to **interpret** functional programs.

This week, we **compile** them to lower-level code.

- This leads to **more efficient** execution. (No interpretive overhead.)
- This helps **understand** advanced language features, such as
  - **first-class functions**,
  - **recursion**,
  - **exceptions**, **effect handlers**, etc.
- This helps understand the organization of memory:
  - **code** versus **data**,
  - the **stack** versus the **heap**.
- This is an occasion to learn some **programming techniques**.
- This is an occasion to use **operational semantics** and discuss **proofs**.

Note: CPS, which explains “recursion” and “the stack”, is optional reading.

## Closure conversion apparatus for existing closure applying machines

U.S. Patent Oct. 20, 1981 Sheet 3 of 10 4,295,320




FIG. 3

Innovation in the carbonated beverage bottling industry is very much dependent on the ready availability of machinery for processing new types of containers and/or **closures**.

This invention provides substitute capping heads to apply a threaded closure on a bottle neck through the utilization of existing machines which were designed to apply an aluminum closure on a bottle neck by in situ formation of the threads in an aluminum shell.

# Motivation

We wish to compile (translate):

- a language with **arbitrary** first-class functions (i.e.,  $\lambda$ -abstractions)

down to:

- a language with **closed** first-class functions (i.e., “code pointers”)

# Motivation

Why might we wish to understand this compilation technique?

- a **key step** in the compilation of functional programming languages;
- a way of **explaining the magic** of first-class functions;
- a way of understanding their **space and time cost**;
- a **programming technique** in languages without first-class functions.

## Passing down a function

A functional programming language, such as OCaml, naturally allows:

- defining **local** (nested) functions,
- passing a function as an **argument** to a function.

This combination of features is sometimes known as “**downward funargs**”.

```
let iter f a =  
  for i = 0 to Array.length a - 1 do f a.(i) done  
let sum a =  
  let s = ref 0 in  
  let add x = (s := !s + x) in  
  iter add a;  
  !s
```

add refers to the variable `s`, which is neither global nor local to `add`.

A **nested** function can refer to a local variable of an **enclosing** function.

## Returning a function

A functional programming language also allows:

- **returning** a function out of a function;
- **storing** a function in a reference, for future use.

A nested function can thus **outlive** a local variable to which it refers.

This is sometimes called “**upward funargs**” (a strange name).

First-class functions have **unbounded** lifetimes;  
therefore, their data cannot be stack-allocated.

## Returning a function

A memory cell that is accessible only via a pair of get and set functions:

```
let make x =  
  let cell = ref x in  
  let get () = !cell  
  and set x = (cell := x) in  
  get, set  
  
let () =  
  let get, set = make 3 in  
  set (get() + 1)
```

A typical example of [procedural abstraction](#) ([Reynolds, 1975](#)), which is widely popular in object-oriented programming languages. On this topic, see also [Cook \(2009\)](#) and [Aldrich \(2013\)](#).

cell is a local variable in make.

It [no longer exists](#) when get and set are called!

How can we [transform](#) this code so as to use only [closed](#) functions? ...



# Principles of closure conversion

## Closure conversion

### Motivation

#### Definition

#### Soundness

#### Type preservat°

#### Digression

## Defun°

## Digression

#### Objects

#### $\lambda$ -lifting

#### SKI

## Conclusion

get and set need access to the **value** of the local variable `cell`.  
(This value is the address of a heap-allocated reference cell.)

- Therefore, they need one more parameter, an **environment**, which somehow gives access to this value;

At a call site, we must be able to supply an environment.

- Therefore, a  $\lambda$ -abstraction must evaluate to a **closure**, which gives access to both the **code** and the **environment**.

A closure must be **heap-allocated**, as its lifetime is unbounded.

## Variations

One must **choose a convention** as to

- the **memory layout** of the closure and environment,
- the **calling convention** of the code.

In the following, I make the following choices:

- the closure and the environment are the same memory block,
- the code expects a pair of the closure and the original argument.

# Closure conversion: example

## Closure conversion

### Motivation

### Definition

### Soundness

### Type preservat<sup>o</sup>

### Digression

## Defun<sup>o</sup>

## Digression

### Objects

### $\lambda$ -lifting

### SKI

## Conclusion

A closure `clo` is a pair of a function code and a pointer cell:

```
let closure code cell = (code, cell) (* builds a closure *)
let project_code = fst (* reads the [code] field *)
let project_cell = snd (* reads the [cell] field *)
```

The code expects a pair of the closure `clo` itself  
and the original function argument `arg`.

```
let apply clo arg =
  let code = project_code clo in
  code (clo, arg)
```

This code is ill-typed in OCaml.

The closure `clo` contains a function code which expects the closure itself  
as an argument. This is not permitted.

## Closure conversion: example

Closure  
conversion

## Motivation

## Definition

## Soundness

Type preservat<sup>o</sup>

## Digression

Defun<sup>o</sup>

## Digression

## Objects

 $\lambda$ -lifting

## SKI

## Conclusion

The result of **closure conversion** is then as follows:

```
let make x =  
  let cell = ref x in  
  let get (clo, ()) = let cell = project_cell clo in !cell  
  and set (clo, x) = let cell = project_cell clo in cell := x  
  in closure get cell, closure set cell  
  
let () =  
  let get, set = make 3 in  
  apply set (apply get () + 1)
```

get and set are now **closed** functions: they have no free variables.

They can be **hoisted** out of make, if desired...

## Closure conversion: example

Closure  
conversion

## Motivation

## Definition

## Soundness

## Type preservat°

## Digression

## Defun°

## Digression

## Objects

## λ-lifting

## SKI

## Conclusion

Here is the result of **hoisting** the closed functions up to the top level:

```
let get (clo, ()) = let cell = project_cell clo in !cell
and set (clo, x) = let cell = project_cell clo in cell := x
```

```
let make x =
  let cell = ref x in
  closure get cell, closure set cell
```

```
let () =
  let get, set = make 3 in
  apply set (apply get () + 1)
```

## Definition of closure conversion

Closure  
conversion

## Motivation

## Definition

## Soundness

## Type preservat°

## Digression

## Defun°

## Digression

## Objects

## λ-lifting

## SKI

## Conclusion

$$\begin{aligned}
 \llbracket x \rrbracket &= x \\
 \llbracket \lambda x. t \rrbracket &= \text{let } code = \lambda(clo, x). \\
 &\quad \text{let } x_i = \pi_i \text{ clo in} \\
 &\quad \llbracket t \rrbracket \\
 &\quad \text{in} \\
 &\quad (code, x_1, \dots, x_n) \\
 \llbracket t_1 \ t_2 \rrbracket &= \text{let } clo = \llbracket t_1 \rrbracket \text{ in} \\
 &\quad \text{let } (code, \dots) = clo \text{ in} \\
 &\quad code \ (clo, \llbracket t_2 \rrbracket)
 \end{aligned}$$

where  $\{x_1, \dots, x_n\} = fv(\lambda x. t)$   
 – note: this  $\lambda$ -abstraction is closed!  
 for each  $i \in \{1, \dots, n\}$

A  $\lambda$ -abstraction with  $n$  free variables becomes a  $(1 + n)$ -tuple.

The target calculus must have tuples. We use the following sugar:

$$\begin{aligned}
 \text{let } (code, \dots) = clo \text{ in } t &\equiv \text{let } code = \pi_0 \text{ clo in } t \\
 \lambda(x, y). t &\equiv \lambda p. \text{let } x = \pi_0 \ p \text{ in let } y = \pi_1 \ p \text{ in } t
 \end{aligned}$$

# Complexity and cost model

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

$\lambda$ -lifting

SKI

Conclusion

Closure conversion leads to the following **cost model**:

- Evaluating a variable costs  $O(1)$ .
- Evaluating a  $\lambda$ -abstraction costs  $O(n)$ ,  
where  $n$  is the number of its free variables.
- Evaluating an application costs  $O(1)$ .

$n$  can be considered  $O(1)$ , as it depends only on the program's text,  
not on the input data.

## Soundness of closure conversion

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

$\lambda$ -lifting

SKI

Conclusion

We would like to state that this program transformation is **sound**.

That is, closure conversion preserves the **meaning** of programs.

We need a **semantic preservation** statement,

roughly:

*If  $t$  exhibits a certain behavior,  
then  $\llbracket t \rrbracket$  exhibits the same behavior.*

How can / should this be stated, more precisely?



## Towards a semantic preservation statement

Closure  
conversion

Motivation  
Definition

**Soundness**  
Type preservat<sup>o</sup>  
Digression

Defun<sup>o</sup>

Digression  
Objects  
 $\lambda$ -lifting  
SKI

Conclusion

To write down such a statement, we must choose:

- a semantics for the source calculus;
- a semantics for the target calculus.

Thoughts?

## Which semantics for the source calculus?

For the call-by-value  $\lambda$ -calculus, we have encountered several semantics:

- small-step, substitution-based;
- big-step, substitution-based;
- big-step, environment-based;
- interpreter, with fuel, environment-based.

They are **equivalent** to one another,  
but perhaps one of them is more **convenient** for this proof?

Let us choose the **big-step, environment-based** semantics.

It **already** has explicit notions of **environments** and **closures**,  
so this choice should make the proof easier.

## Which semantics for the target calculus?

The target of closure conversion is also a  $\lambda$ -calculus.

We **could** therefore use the same semantics for it...?

The statement of semantic preservation would then be:

$$\text{If } e \vdash t \downarrow_{cbv} c, \text{ then } \llbracket e \rrbracket \vdash \llbracket t \rrbracket \downarrow_{cbv} \llbracket c \rrbracket.$$

where the **same semantics** appears in the hypothesis and conclusion.

We **could** prove such a theorem, but...

That would be philosophically and technically **unsatisfactory**:

- the **identity transformation** would satisfy this statement, too!
- this statement does **not** explain in what way closure conversion is a **useful** step in a compiler pipeline. It should lead to a **simpler** calculus!

## Which semantics for the target calculus?

Something is **special** about the target calculus,

or about the **subset** of the target calculus that we exploit.

Every  $\lambda$ -abstraction is **closed** – that is the **point** of closure conversion!

We **should** therefore equip this calculus with a **simplified** semantics which does **not** involve any closures.

## Which semantics for the target calculus?

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

Recall the **standard** big-step, environment-based semantics:

$$\frac{\text{EBigCBVVar} \quad e(x) = c}{e \vdash x \downarrow_{cbv} c}$$

$$\frac{\text{EBigCBVLam} \quad fv(\lambda x.t) \subseteq dom(e)}{e \vdash \lambda x.t \downarrow_{cbv} \langle \lambda x.t \mid e \rangle}$$

$$\frac{\begin{array}{l} \text{EBigCBVApp} \\ e \vdash t_1 \downarrow_{cbv} \langle \lambda x.u_1 \mid e' \rangle \\ e \vdash t_2 \downarrow_{cbv} c_2 \\ e'[x \mapsto c_2] \vdash u_1 \downarrow_{cbv} c \end{array}}{e \vdash t_1 t_2 \downarrow_{cbv} c}$$

where  $c ::= \langle \lambda x.t \mid e \rangle$  and  $e ::= [] \mid e[x \mapsto c]$ .

How can this semantics be **simplified**?

## A semantics for the target calculus

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

No **closures**! Raw  $\lambda$ -abstractions instead. Think of them as **code pointers**.

$$\text{MBigCBVVAR} \quad \frac{e(x) = c}{e \vdash x \Downarrow_{cbv} c}$$

$$\text{MBigCBVLAM} \quad \frac{fv(\lambda x.t) \subseteq \emptyset}{e \vdash \lambda x.t \Downarrow_{cbv} \lambda x.t}$$

$$\text{MBigCBVAPP} \quad \frac{\begin{array}{l} e \vdash t_1 \Downarrow_{cbv} \lambda x.u_1 \\ e \vdash t_2 \Downarrow_{cbv} c_2 \\ [] [x \mapsto c_2] \vdash u_1 \Downarrow_{cbv} c \end{array}}{e \vdash t_1 t_2 \Downarrow_{cbv} c}$$

where  $c ::= \lambda x.t$  and  $e ::= [] \mid e[x \mapsto c]$ .

Such a semantics explains in what way the target calculus is **simpler**.

**Exercise:** extend this semantics with “*let*” constructs, pairs and projections. The solution is in **MetalBigStep**.

## Towards a semantic preservation statement

The semantic preservation statement should be, roughly:

*If  $e \vdash t \downarrow_{cbv} c$ , then  $\llbracket e \rrbracket \vdash \llbracket t \rrbracket \Downarrow_{cbv} \llbracket c \rrbracket$ .*

What is **missing still** for this statement to make sense?

We must **define** the translation of environments  $\llbracket e \rrbracket$  and closures  $\llbracket c \rrbracket$ .

An **environment** is naturally transformed pointwise:

$$\begin{aligned} \llbracket [] \rrbracket &= [] \\ \llbracket e[x \mapsto c] \rrbracket &= e[x \mapsto \llbracket c \rrbracket] \end{aligned}$$

A **closure** is represented as... **a tuple**:

$$\begin{aligned} \llbracket \langle \lambda x.t \mid e \rangle \rrbracket &= (\lambda(clo, x). \dots, \llbracket e(x_1) \rrbracket, \dots, \llbracket e(x_n) \rrbracket) \\ &\text{where } \{x_1, \dots, x_n\} = fv(\lambda x.t) \\ &\text{and } x_1 < \dots < x_n \quad \text{— a **deterministic** convention} \end{aligned}$$

## A semantic preservation statement

Semantic preservation is stated as follows:

### Lemma (Semantic preservation)

*Assume  $e \vdash t \downarrow_{cbv} c$ .*

*Then,  $\llbracket e \rrbracket \vdash \llbracket t \rrbracket \Downarrow_{cbv} \llbracket c \rrbracket$  holds.*

**Exercise** (recommended): write a careful proof of this lemma, on paper.



## A semantic preservation statement

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

In Rocq, we use de Bruijn style, and prove:

### Lemma (Semantic preservation)

Assume  $e \vdash t \downarrow_{cbv} c$ .

Then,  $\llbracket e \rrbracket \vdash \llbracket t \rrbracket_n \Downarrow_{cbv} \llbracket c \rrbracket$  holds,  
where  $n$  is the length of the environment  $e$ .

### Proof.

By induction on the hypothesis.

See [ClosureConversion/semantic\\_preservation](#). □

Here, the transformation  $\llbracket \cdot \rrbracket$  takes **two** parameters  $t$  and  $n$ ,  
where the free variables of  $t$  are expected to be below  $n$ .

**Exercise** (tricky): define  $\llbracket t \rrbracket_n$  and prove that if  $t$  has free variables below  $n$   
then  $\llbracket t \rrbracket_n$  has free variables below  $n$ . See [ClosureConversion/fv\\_cc](#).

Are we happy with this semantic preservation statement?

$$e \vdash t \downarrow_{cbv} c \text{ implies } \llbracket e \rrbracket \vdash \llbracket t \rrbracket \Downarrow_{cbv} \llbracket c \rrbracket.$$

This is a **forward** preservation statement.

Does it **actually mean** what we think it means?

What do we **think** it means? Perhaps this?

*The transformed program “**computes the same thing**”  
as the source program.*

Or this?

*The transformed program “**behaves in the same way**”  
as the source program.*

## Caveat 1: is the translation nontrivial?

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>n</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

 $\lambda$ -lifting

SKI

Conclusion

The trivial transformation defined by  $\llbracket t \rrbracket = ()$   
also satisfies this semantic preservation statement.

We should somehow check that our transformation is nontrivial.

Some concrete observation of program behavior must be preserved:

- If our calculi had primitive integers, we could check that  $\llbracket i \rrbracket$  is  $i$ , so a program that computes an integer value is transformed to a program that computes the same value.
- We could also check that  $\llbracket t \rrbracket$  terminates if and only if  $t$  terminates. This brings us to the next slide...

## Caveat 2: is nontermination preserved?

If  $t$  diverges (i.e., does not terminate), then “obviously”  $\llbracket t \rrbracket$  diverges too.

However, we have **not** proved this fact:

it does **not** follow from our semantic preservation statement.

A transformation that maps certain divergent programs to  $()$   
**would still satisfy** this statement!

To prove that nontermination is preserved, we could use:

- a small-step semantics, or
- a **co-inductive** nontermination judgement  $e \vdash t \uparrow_{\text{cbv}}$ .  
See **Leroy and Grall (2007)**.

## Caveat 3: forward versus backward preservation

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

$\lambda$ -lifting

SKI

Conclusion

We have proved:

*Every behavior of the source program  
is also a behavior of the transformed program.*

That is,

*The behaviors of the source program  
form a **subset** of the behaviors of the transformed program.*

So,

*A program that prints “hello”  
can be transformed to  
a program that prints “hello” **OR** launches a missile.*



## Caveat 3: forward vs. backward preservation

We really need **backward** preservation:

*The behaviors of the source program  
form a **superset**  
of the behaviors of the transformed program.*

## Caveat 3: forward versus backward preservation

If the target calculus has **deterministic** semantics,  
then the transformed program has exactly one behavior,  
so forward and backward preservation coincide.

If the target calculus has **nondeterministic** semantics,  
then backward preservation must be proved directly.

- A naïve approach may be painful.
- **Relational Separation Logic** is a tool that helps do this.

Yang, **Relational Separation Logic**, 2007.

Frumin, Krebbers, Birkedal, **ReLoC Reloaded**, 2021.

Gäher et al., **Simuliris: a separation logic framework for verifying  
concurrent program optimizations**, 2022.

Allain, Clément, Pottier, Scherer, **Tail Modulo Cons, OCaml, and  
Relational Separation Logic**, 2025.

## Caveats – summary

In summary, **always question what you have proved.**

Even if your proofs are machine-checked,  
your definitions and statements can be incorrect or deceiving.



# Understanding programs through closure conversion

People sometimes use first-class functions in somewhat mysterious ways.

Closure conversion can **help understand** what is going on.

Let us look at a little example: **difference lists**.

This also illustrates **type-preserving** closure conversion in OCaml.

## A fringe computation

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

λ-lifting

SKI

Conclusion

Suppose we have a type of trees with integer-labeled leaves:

```
type tree =  
  | Leaf of int  
  | Node of tree * tree
```

Suppose we wish to construct a tree's **fringe**, a sequence of integers:

```
let rec fringe (t : tree) : int list =  
  match t with  
  | Leaf i          -> [ i ]  
  | Node (t1, t2) -> fringe t1 @ fringe t2
```

What do you think?

This code is inefficient. Its worst-case time complexity is **quadratic**.

## Difference lists

To remedy this, people sometimes use **difference lists** (Hughes, 1984).

```
type 'a diff =  
  'a list -> 'a list  
  
let singleton (x : 'a) : 'a diff =  
  fun xs -> x :: xs  
let concat (xs : 'a diff) (ys : 'a diff) : 'a diff =  
  fun zs -> xs (ys zs)
```

The idea is to represent a list  $xs$  as a function that maps  $ys$  to  $xs @ ys$ .

`singleton` is “cons”, and `concat` is function composition!

They both have time complexity  $O(1)$ .

With difference lists, the fringe computation is as follows:

```
let rec fringe_ (t : tree) : int diff =  
  match t with  
  | Leaf i      -> singleton i  
  | Node (t1, t2) -> concat (fringe_ t1) (fringe_ t2)  
  
let fringe t =  
  fringe_ t []
```

Is this code efficient? What is its time complexity?

The application `fringe_ t` costs  $O(n)$ ,  
and the application of its result to the empty list `[]` costs  $O(n)$  as well.

That is somewhat nonobvious, though. [Closure conversion](#) to the rescue!

## A type of all closures

A type of closures can be defined in OCaml as follows:

```
type ('a, 'b) closure =  
  | Clo:  
      ?                               (* A (closed) function... *)  
      * ?                             (* ...and its environment... *)  
  -> ('a, 'b) closure (* ...together form a closure. *)
```

```
type ('a, 'b) closure =  
  | Clo:  
      ('a * 'e -> 'b) (* A (closed) function... *)  
      * 'e            (* ...and its environment... *)  
  -> ('a, 'b) closure (* ...together form a closure. *)
```

This is an **existential type**:  $(\alpha, \beta) \text{ closure} \simeq \exists \epsilon. ((\alpha \times \epsilon) \rightarrow \beta) \times \epsilon$ .

Here, the closure and its environment are **distinct** memory blocks; this is slightly different from what was shown earlier today.

To read more about type-preserving closure conversion, see [Minamide et al. \(1996\)](#) and my ancient [slides \(2009\)](#).

## Well-typed closure invocation

Because all closures have the same type, the code that invokes a closure can be written **once and for all**, and is polymorphic.

```
let apply (f : ('a, 'b) closure) (x : 'a) : 'b =  
  let Clo (code, env) = f in  
  code (x, env)
```

env has unknown type, yet the call `code (x, env)` is definitely safe.

## Difference lists, closure-converted

Let us manually apply closure conversion to difference lists.

A difference list is a closure:

```
type 'a diff =  
  ('a list, 'a list) closure
```

singleton and concat allocate and return a closure:

```
let singleton_code =  
  fun (xs, x) -> x :: xs  
let singleton (x : 'a) : 'a diff =  
  Clo (singleton_code, x)  
  
let concat_code =  
  fun (zs, (xs, ys)) -> apply xs (apply ys zs)  
let concat (xs : 'a diff) (ys : 'a diff) : 'a diff =  
  Clo (concat_code, (xs, ys))
```

The **closed** functions `singleton_code` and `concat_code` are **hoisted out**.

## Difference lists, closure-converted

The difference-list-based fringe computation, closure-converted:

```
let rec fringe_ (t : tree) : int diff =  
  match t with  
  | Leaf i      -> singleton i  
  | Node (t1, t2) -> concat (fringe_ t1) (fringe_ t2)  
  
let fringe t =  
  apply (fringe_ t) []
```

It should be clear that `fringe_ t` builds a [tree of closures](#), whose shape is identical to the shape of the tree `t`.

Then, the call `apply _ []` interprets this tree of closures.

Is this efficient?

Each of the two phases costs  $O(n)$ . Asymptotic complexity is good.

Yet, phase 1, a tree copy, is [useless](#). The constant factor is not great.

[Exercise](#): write `fringe` so that cost is  $O(n)$  and no tree copy takes place.



## Minimal environments

My pencil-and-paper definition of closure conversion allocates closures with **minimal environments**, where the closure stores **just what it needs**:

$$\begin{aligned} \llbracket \lambda x.t \rrbracket = & \text{let } code = & \text{where } \{x_1, \dots, x_n\} = fv(\lambda x.t) \\ & \lambda(clo, x). \\ & \text{let } x_i = \pi_i \text{ } clo \text{ in} & \text{for each } i \in \{1, \dots, n\} \\ & \llbracket t \rrbracket \\ & \text{in} \\ & (code, x_1, \dots, x_n) \end{aligned}$$

In contrast, for greater simplicity, my Rocq definition in **ClosureConversion** allocates closures with **full environments**.

Full environments lead to larger closures and **space leaks**, so they are not viable in practice.

## Recursive functions

How should **recursive function definitions** be transformed?

$$\begin{aligned} \llbracket \lambda x. t \rrbracket &= \text{let } \text{code} = && \text{where } \{x_1, \dots, x_n\} = \text{fv}(\lambda x. t) \\ &\lambda(\text{clo}, x). && \\ &\text{let } x_i = \pi_i \text{ clo in} && \text{for each } i \in \{1, \dots, n\} \\ &\llbracket t \rrbracket && \\ &\text{in} && \\ &(\text{code}, x_1, \dots, x_n) && \end{aligned}$$

$$\llbracket \mu f. \lambda x. t \rrbracket = ?$$

$$\begin{aligned} \llbracket t_1 \ t_2 \rrbracket &= \text{let } \text{clo} = \llbracket t_1 \rrbracket \text{ in} \\ &\text{let } (\text{code}, \dots) = \text{clo in} \\ &\text{code } (\text{clo}, \llbracket t_2 \rrbracket) \end{aligned}$$

The transformation of **function calls** cannot be modified: the caller does not (want to) know whether the callee is a recursive function.

In other words, we must maintain a **uniform calling convention**.

# Recursive functions

We must arrange for the variable  $f$  to be bound to an appropriate value:

$$\begin{aligned} \llbracket \mu f. \lambda x. t \rrbracket = & \text{let } code = & \text{where } \{x_1, \dots, x_n\} = fv(\mu f. \lambda x. t) \\ & \lambda(clo, x). & \\ & \text{let } f = clo \text{ in} & \\ & \text{let } x_i = \pi_i \text{ } clo \text{ in} & \text{for each } i \in \{1, \dots, n\} \\ & \llbracket t \rrbracket & \\ & \text{in} & \\ & (code, x_1, \dots, x_n) & \end{aligned}$$

The closure  $clo$  happens to be such a value.

**Exercise:** extend the Rocq formalisation with recursive functions.

# Mutually recursive functions

What about **mutually recursive** function definitions?

Suppose the source calculus has mutually recursive **function** definitions:

$$t ::= \dots \mid \text{let rec } f_1 = \lambda x_1. t_1 \text{ and } f_2 = \lambda x_2. t_2 \text{ in } t$$

Suppose the target calculus has mutually recursive **value** definitions:

$$t ::= \dots \mid \text{let rec } f_1 = v_1 \text{ and } f_2 = v_2 \text{ in } t$$

**Exercise:** propose a (deterministic, call-by-value) operational semantics with support for these constructs. Impose a syntactic **contractiveness** condition to rule out nonsensical definitions such as **let rec**  $x = x$ .

**Answer:** Nordlander, Carlsson and Gill (2008) propose an operational semantics. Reynaud, Scherer and Yallop (2019) propose a type system that rules out nonsensical definitions.

## Mutually recursive functions

It is then easy to define a transformation of recursive functions:

$$\begin{aligned} \llbracket \text{let rec } f_1 = \lambda x_1. t_1 \\ \text{and } f_2 = \lambda x_2. t_2 \\ \text{in } u \rrbracket = \text{let rec } f_1 = \llbracket \lambda x_1. t_1 \rrbracket \\ \text{and } f_2 = \llbracket \lambda x_2. t_2 \rrbracket \\ \text{in } \llbracket u \rrbracket \end{aligned}$$

This transformation is correct, albeit slightly inefficient.

We get **two closures**, each of which contains  
**a pointer to itself** and  
**a pointer to the other closure**.

Indeed,  $f_1$  and  $f_2$  can be free in  $\lambda x_1. t_1$  and  $\lambda x_2. t_2$ .

If so, the tuples  $\llbracket \lambda x_1. t_1 \rrbracket$  and  $\llbracket \lambda x_2. t_2 \rrbracket$  have slots for  $f_1$  and  $f_2$ .

## Mutually recursive functions

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

A slightly more efficient transformation is as follows:

$$\begin{aligned} \llbracket \text{let rec } f_1 &= \lambda x_1. t_1 \\ &\text{and } f_2 = \lambda x_2. t_2 \\ &\text{in } u \rrbracket = \text{let rec } f_1 = \llbracket \mu f_1. \lambda x_1. t_1 \rrbracket \\ &\text{and } f_2 = \llbracket \mu f_2. \lambda x_2. t_2 \rrbracket \\ &\text{in } \llbracket u \rrbracket \end{aligned}$$

We get **two closures**, each of which points to the other.

The redundant pointer from each closure to itself has been removed.

## Some history

Closures were used in interpreters in the 1960s and 1970s.

Landin, *The Mechanical Evaluation of Expressions*, 1964.

Sussman and Steele,

*SCHEME: an interpreter for extended lambda-calculus*, 1975.

Closure conversion in a compiler (first?) appeared in Rabbit.

Steele, *RABBIT: a compiler for SCHEME*, 1978.

## Further improvements

Naïve closure conversion can produce inefficient code.

**Selective** closure conversion applies when the environment would have zero slots, and avoids building a closure in that case.

**Lightweight** closure conversion applies when a value that should be stored in the environment happens to be available at every call site. Then, instead of being stored, this value becomes an extra argument.

Both ideas involve a **modified closure invocation protocol**, therefore require an **agreement** between callers and callees, therefore require a **control flow analysis**: one must know which closures may be invoked where.

Steckler and Wand, **Selective and lightweight closure conversion**, 1994.

Cejtin *et al.*, **Flow-directed closure conversion for typed languages**, 2000.  
— actually, defunctionalization



## Further improvements

Applications of (curried) **multi-parameter functions** to multiple actual arguments should be identified and optimized.

- e.g., `map f xs` should **not** be compiled as `(map f) xs`, where `(map f)` allocates and returns a closure!

This is done in OCaml and in CakeML, where this transformation is **verified**.

Owens *et al.*, **Verifying Efficient Function Calls in CakeML**, 2017.

## Trivia: closures and mutable variables

In ML, F#, Java, ..., closures capture **immutable** variables only.


In certain “interesting” programming languages, closures can refer to **mutable** variables, too. E.g., in JavaScript:

```
var messages = ["Wow!", "Hi!", "Closures are fun!"];  
for (var i = 0; i < messages.length; i++) {  
  setTimeout(function () {  
    say(messages[i]);  
  }, i * 1500);  
}
```

This program, of course, ... prints undefined three times.

See [Orendorff's blog post](#) for details.

## Activation of TRPV1 by capsaicin



Activation of TRPV1 by capsaicin results in sensory neuronal depolarization, and can induce local sensitization to activation by heat, acidosis, and endogenous agonists. Topical exposure to capsaicin leads to the sensations of heat, burning, stinging, or itching. High concentrations of capsaicin or repeated applications can produce a persistent local effect on cutaneous nociceptors, which is best described as **defunctionalization** and constituted by reduced spontaneous activity and a loss of responsiveness to a wide range of sensory stimuli.

# Defunctionalization

Defun<sup>o</sup>, like closure conversion, aims to **eliminate first-class functions**.

Defun<sup>o</sup> differs in the representation of closures:

- instead of a **code pointer** and an environment,
- use a **tag** and an environment.

Thus, instead of (closed) first-class functions, the target language must have **algebraic data types**.

Functions become data!

## Definition of defunctionalization

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

 $\lambda$ -lifting

SKI

Conclusion

Assume that every  $\lambda$ -abstraction is decorated with a **distinct** label  $C$ .

$$\llbracket x \rrbracket = x$$

$$\llbracket \lambda^C x. t \rrbracket = C(x_1, \dots, x_n) \quad \text{where } \{x_1, \dots, x_n\} = fv(\lambda x. t)$$

$$\llbracket t_1 \ t_2 \rrbracket = \textit{apply} (\llbracket t_1 \rrbracket, \llbracket t_2 \rrbracket)$$

Simple, eh?

There remains to define *apply*.

## Definition of defunctionalization

The toplevel function *apply* interprets a closure as a function.

The whole transformed program is placed in the scope of this definition:

*let rec apply (this, that) =*

*match this with*

*...*

|  $C(x_1, \dots, x_n) \rightarrow$

*let x = that in*

$\llbracket t \rrbracket$

*...*

for each function  $\lambda^C x. t$  in the source

*apply* must be recursively defined, as  $\llbracket t \rrbracket$  can refer to *apply*.

# Soundness of defunctionalization

Is defunctionalization sound, that is, **semantics-preserving**?

Of course it is!

For a (paper) proof in an untyped setting, see **Pottier and Gauthier (2006)**.

## Difference lists

Recall difference lists:

```
type 'a diff =  
  'a list -> 'a list  
  
let singleton (x : 'a) : 'a diff =  
  fun xs -> x :: xs  
  
let concat (xs : 'a diff) (ys : 'a diff) : 'a diff =  
  fun zs -> xs (ys zs)
```

As an illustration, let us manually defunctionalize this code.

There are **two** first-class functions of interest in this code.



## Difference lists, defunctionalized

Thus, we introduce an algebraic data type with **two** data constructors:

```
type 'a diff =  
  | Singleton of 'a  
  | Concat of 'a diff * 'a diff  
let singleton (x : 'a) : 'a diff =  
  Singleton x  
let concat (xs : 'a diff) (ys : 'a diff) : 'a diff =  
  Concat (xs, ys)
```

The functions `singleton` and `concat` just build a closure.

There remains to define `apply`.

## Difference lists, defunctionalized

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

 $\lambda$ -lifting

SKI

Conclusion

apply applies a closure this to an argument that.

```
let rec apply (this : 'a diff) (that : 'a list) : 'a list =  
  match this with  
  | Singleton x ->  
    let xs = that in  
    x :: xs  
  | Concat (xs, ys) ->  
    let zs = that in  
    apply xs (apply ys zs)
```

## Difference lists, defunctionalized

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

The fringe computation is the same as in the closure-converted version:

```
let rec fringe_ (t : tree) : int diff =  
  match t with  
  | Leaf i      -> singleton i  
  | Node (t1, t2) -> concat (fringe_ t1) (fringe_ t2)  
  
let fringe t =  
  apply (fringe_ t) []
```

The types `tree` and `int diff` are *isomorphic*!

It is clear (again) that `fringe_` is just a tree copy.

# Sets as characteristic functions

Let us use defunctionalization to investigate another slightly mysterious piece of code, namely **sets** implemented as **functions**.

# Sets as characteristic functions

A set (of integers) can be represented by its characteristic function:

```
type set =  
  int -> bool  
let empty : set =  
  fun y -> false  
let singleton (x : int) : set =  
  fun y -> y = x  
let union (s1 : set) (s2 : set) : set =  
  fun y -> s1 y || s2 y  
let mem (x : int) (s : set) : bool =  
  s x
```

This works. But is it smart... or is it not?

empty, singleton, union have time complexity  $O(1)$ . What about mem?

Answering requires understanding the structure of the closure  $s$ .

# Sets as characteristic functions

There are **three places** where a closure of type set is built:

```
let empty : set =  
  fun y -> false  
let singleton (x : int) : set =  
  fun y -> y = x  
let union (s1 : set) (s2 : set) : set =  
  fun y -> s1 y || s2 y
```

What fields do these closures carry?

- in empty, **no fields**;
- in singleton, **one field** of type **int** corresponding to x;
- in union, **two fields** of type set corresponding to s1 and s2.

Let us give these three kinds of closures three distinct labels, say **E**, **S**, **U**.

Sets as characteristic functions,  
defunctionalizedClosure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

A “set” is really a closure of one of these three kinds.

Through defunctionalization, set becomes an algebraic data type:

```
type set =  
  | E  
  | S of int  
  | U of set * set
```

The three constructor functions become:

```
let empty : set = E  
let singleton (x : int) : set = S (x)  
let union (s1 : set) (s2 : set) : set = U (s1, s2)
```

What is this data type?

A data type of trees with leaves **E** and **S** and binary nodes **U**.

Sets as characteristic functions,  
defunctionalizedClosure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

λ-lifting

SKI

Conclusion

apply interprets a set as a characteristic function of type `int -> bool`.

```
let rec apply (s : set) (y : int) : bool =  
  match s with  
  | E          -> false  
  | S (x)      -> y = x  
  | U (s1, s2) -> apply s1 y || apply s2 y
```

The membership test becomes:

```
let mem (x : int) (s : set) : bool =  
  apply s x
```



Defun° helps see that a set is represented as an **unbalanced** tree.

mem s x **traverses** all of the tree s in search of x.

mem has time complexity  $O(n)$ . **Inefficient!**

Understanding closure conversion and/or defun° helps analyze this code.

## Type-preserving defunctionalization

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

$\lambda$ -lifting

SKI

Conclusion

We have seen earlier that closure conversion can be **type-preserving**. This requires **existential types** in the target calculus.

Can defunctionalization be made type-preserving?

**Yes**, it can. This requires **generalized algebraic data types** (GADTs) in the target calculus.

Pottier and Gauthier,  
**Polymorphic typed defunctionalization and concretization**, 2006.

# Type-preserving defunctionalization

Why are GADTs needed? Well, try defunctionalizing this OCaml code:

```
let foo (xs : int list) : int list =  
  filter (fun x -> x <> 0) (map (fun x -> x + 1) xs)
```

The solution appears on the following slides.

## Type-preserving defunctionalization

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

```
let foo (xs : int list) : int list =  
  filter (fun x -> x <> 0) (map (fun x -> x + 1) xs)
```

A GADT is needed to describe the two kinds of closures that appear:

```
type (_, _) closure =  
  | NEqZero    : (int, bool) closure  
  | Increment  : (int, int) closure
```

These closures are trivial: there is just a tag, no fields.

## Type-preserving defunctionalization

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

```
let foo (xs : int list) : int list =  
  filter (fun x -> x <> 0) (map (fun x -> x + 1) xs)
```

These closures are interpreted by the function apply:

```
let rec apply =  
  fun (type a b) (this : (a, b) closure) (that : a) : b ->  
    match this with  
    | NEqZero ->  
      let x = that in  
      x <> 0  
    | Increment ->  
      let x = that in  
      x + 1
```

## Type-preserving defunctionalization

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

```
let foo (xs : int list) : int list =  
  filter (fun x -> x <> 0) (map (fun x -> x + 1) xs)
```

The functions `filter` and `map` use `apply`:

```
let rec filter p xs =  
  match xs with  
  | [] -> []  
  | x :: xs -> if apply p x then x :: filter p xs  
               else filter p xs  
  
let rec map f xs =  
  match xs with  
  | [] -> []  
  | x :: xs -> apply f x :: map f xs
```

## Type-preserving defunctionalization

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

 $\lambda$ -lifting

SKI

Conclusion

```
let foo (xs : int list) : int list =  
  filter (fun x -> x <> 0) (map (fun x -> x + 1) xs)
```

The  $\lambda$ -abstractions are transformed into closures:

```
let foo (xs : int list) : int list =  
  filter NEqZero (map Increment xs)
```

## Some history

Defunctionalization is applied by Reynolds to an interpreter.

More about this in the lecture on CPS.

Reynolds, **Definitional interpreters  
for programming languages**, 1972 (1998).

Reynolds, **Definitional interpreters revisited**, 1998.

Defunctionalization is used in some compilers, e.g., **MLton**.

They use **data flow analysis** to create multiple **specialized** apply functions,  
which dispatch on fewer cases.

Cejtin, Jagannathan, Weeks,  
**Flow-directed Closure Conversion for Typed Languages**, 2000.



# Compiling functions as objects

On the surface, Scala has functions **and** objects, but functions are **sugar**.

The function type  $T \Rightarrow R$  is sugar for `Function1[T, R]`.

```
trait Function1[-T, +R] { // Defined in the base library.  
  def apply(v: T): R      // Any object with an apply method  
                           // is a "function!"  
}
```

An anonymous function:

```
(x: Int) => x + y
```

is sugar for an object creation expression:

```
new Function1[Int, Int] {  
  def apply(x: Int): Int = x + y  
}
```

Functions (w/ free variables) are translated to objects (w/ free variables).

# Compiling objects with free variables

This object creation expression has a free variable `y`:

```
new Function1[Int, Int] {  
  def apply(x: Int): Int = x + y  
}
```

It can be viewed as sugar for

```
new C (y)
```

where `C` is a unique name. The class `C` is defined at the top level:

```
class C (y: Int) {  
  def apply(x: Int): Int = x + y  
}
```

Objects (w/ free variables) are translated to [parameterized](#) classes.

This is a [modular](#) form of defunctionalization.

## Compiling parameterized classes

A parameterized class:

```
class C (y: Int) {  
  def apply(x: Int): Int = x + y  
}
```

is sugar for a class with an explicit field and a constructor:

```
class C {  
  var y: Int = _  
  def this (y: Int) = {  
    this()  
    this.y = y  
  }  
  def apply(x: Int): Int = x + y  
}
```

Parameterized classes are translated down to ordinary classes.

# Functions in Scala

Thus, functions in Scala are represented by **heap-allocated closures**, with one field per free variable, just as in OCaml.

First-class functions (with free variables) can also be compiled down to a combination of closed (toplevel) functions and **partial applications**.

This is known as  $\lambda$ -lifting.

Johnsson, **Lambda lifting:  
transforming programs to recursive equations**, 1985.

eval evaluates a polynomial cs at a point x.

```
let eval (cs : int list) (x : int) : int =  
  let cons (c : int) (a : int -> int) =  
    let aux x_n = c * x_n + a (x * x_n) in  
    aux  
  and null x_n =  
    0  
  in foldr cons null cs 1
```

aux has three free variables c, a, x, while cons has one free variable x.

Step 1: make them **parameters**.

Every function in the resulting code is **closed**.

```
let eval cs x =  
  let cons x c a =  
    let aux c a x x_n = c * x_n + a (x * x_n) in  
    aux c a x  
  and null x_n =  
    0  
in foldr (cons x) null cs 1
```

Step 2: **hoist** every function to the top level.

We get a group of toplevel functions, also known as [supercombinators](#):

```
let aux c a x x_n =  
  c * x_n + a (x * x_n)  
let cons x c a =  
  aux c a x  
let null x_n =  
  0  
let eval cs x =  
  foldr (cons x) null cs 1
```

This code contains [partial applications](#), though. (Find them!)



$\lambda$ -lifting was popular in the 1980s because people were interested in abstract machines or “graph reduction” systems that dealt with partial applications directly.

Peyton Jones, *The implementation of functional programming languages*, 1987.

Bird, *A formal development of an efficient supercombinator compiler*, 1987.

Today, these ideas are largely *obsolete*.

## Compiling to SKI combinators

Closure  
conversion

Motivation

Definition

Soundness

Type preservat°

Digression

Defun°

Digression

Objects

λ-lifting

SKI

Conclusion

An idea from the 1950s (Curry and Feys, 1958).

The combinators  $S$ ,  $K$ ,  $I$  are defined as follows:

$$\begin{aligned} S\ f\ g\ x &= f\ x\ (g\ x) \\ K\ x\ y &= x \\ I\ x &= x \end{aligned}$$

What is special about them?

Every  $\lambda$ -term can be compiled to code that involves just  $S$ ,  $K$ ,  $I$  and applications.

No  $\lambda$ -abstractions, no variables!

## Compiling to SKI combinators

Apply the following rules, beginning with innermost  $\lambda$ -abstractions:

replace  $\lambda x.x$  with  $I$   
replace  $\lambda x.t$  with  $K t$  if  $x \notin fv(t)$   
replace  $\lambda x.(t_1 t_2)$  with  $S (\lambda x.t_1) (\lambda x.t_2)$

Thus,  $\lambda$ 's are pushed towards the leaves, where they vanish.

For instance,

$$\begin{aligned} & (\lambda x.+ x x) 5 \\ \rightarrow & S (\lambda x.+ x) (\lambda x.x) 5 \\ \rightarrow & S (S (\lambda x.+ x) (\lambda x.x)) (\lambda x.x) 5 \\ \rightarrow & S (S (K +) (\lambda x.x)) (\lambda x.x) 5 \\ \rightarrow & S (S (K +) I) (\lambda x.x) 5 \\ \rightarrow & S (S (K +) I) I 5 \end{aligned}$$

Recursion can be dealt with by adding the combinator  $Y$ .

This naïve approach can cause a quadratic blowup, though.

Kiselyov,  $\lambda$  to SKI, Semantically, 2018.

## Compiling to SKI combinators

Closure  
conversion

Motivation

Definition

Soundness

Type preservat<sup>o</sup>

Digression

Defun<sup>o</sup>

Digression

Objects

$\lambda$ -lifting

SKI

Conclusion

$S$ ,  $K$ ,  $I$  can be viewed as the instruction set of an abstract machine.

This idea was used to compile SASL and Miranda (Turner, 1976–79).

There were plans in the 1980s to do this in hardware!

Peyton Jones, **The implementation  
of functional programming languages**, 1987.

Of course, it is much more efficient to compile down to machine code for a standard (von Neumann) processor, which is highly optimized.

## A few things to remember

- First-class functions are a **key feature** of modern prog. languages.
- Closure conversion and defunctionalization **help understand** them.
- Like objects, closures bundle **code** and **data**  
— “**behavior**” and “**state**”, in object-oriented speak.
- We have encountered a **semantic preservation** statement and discussed its exact meaning. Always **question** what you have proved!